http://www.mapleprimes.com/questions/147724-Fsolve-Inside-An-Integral

```
> restart; interface(version); Digits:=15;
  'Int(RootOf(Int(k*r*PDF(R,r),r=-infinity..x)=0,x)*PDF(Q,z),z=0..1) = ` ?`';
```

Classic Worksheet Interface, Maple 17.00, Windows, Feb 21 2013, Build ID 813473

$$Digits := 15$$

$$\int_0^1 \mathrm{RootOf}\left(\int_{-\infty}^{x} k\, r\, \mathrm{PDF}(R, r)\, dr = 0, x\right) \mathrm{PDF}(Q, z)\, dz = \ ?$$

```
> with(Statistics):
> dummy := sqrt((1+y*e)/(y*e)):
  dummy1 := sqrt(1/(1+y*e)):
  Q := RandomVariable(Normal(0, dummy)):
  R := RandomVariable(Normal(y*e*z/(1+y*e), dummy1)):
  y := 1:
  e := 1/5: #.2;
  k := 1:
  #z :=1;
  l:= 1/2: #0.5;
  a:= 5/4: #1.25;
```

The inner integral has a formal solution, write it in a 'nice' way

```
> 'JJ'='Int(k*r*PDF(R,r),r=-infinity..x)';
  value(rhs(%)):
  simplify(expand(%), size): combine(%, exp):
  Student[Precalculus]:-CompleteSquare(%,x):
  collect(%, Pi): simplify(%, size): collect(%, Pi):
  collect(%, z):
  JJ:=unapply(%,x,z);
```

$$JJ = \int_{-\infty}^{x} k\, r\, \mathrm{Statistics:\text{-}PDF}(R, r)\, dr$$

$$JJ := (x, z) \rightarrow \left(\frac{1}{12}\mathrm{erf}\left(\frac{1}{30}\sqrt{5}\sqrt{3}\,(6\,x - z)\right) + \frac{1}{12}\right) z - \frac{1}{6}\,\frac{\sqrt{5}\sqrt{3}\,\mathbf{e}^{(-1/60\,(6\,x - z)^2)}}{\sqrt{\pi}}$$

Carl Love shows, how to use 'fsolve' to find zeros of JJ and how to find the demanded value:

```
> OIF:= z -> fsolve(JJ(x,z), x);
```

$$OIF := z \rightarrow \mathrm{fsolve}(JJ(x, z), x)$$

```
> st:=time():
  'Int(z-> OIF(z)*PDF(Q,z), 0 .. 1)';
  ``=evalf(%);
  `time used`[sec] = time() - st;
```

$$\mathrm{Int}(z \rightarrow \mathrm{OIF}(z)\ \mathrm{Statistics:\text{-}PDF}(Q, z), 0 .. 1)$$

$$= 0.282975811112918$$

$$\text{time used}_{sec} = 18.517$$

One can improve timings as follows:
1. Find a good initial guess for solving for the zeros.
2. Find a representation for finding zeros which behaves well at the singularity z=0.

Rewrite the task to find zeros in terms of the standard normal (after starring at the solution JJ)

```
> RandomVariable(Normal(0, 1)): CDF(%, t):
  N:=unapply(%,t);
  ``;
  'JJ(x,z)'='eval((N(t)*z - D(N)(t)*sqrt(30)), t = (6*x-z)/sqrt(30))/6';
  expand(%):
  is(%);
```

$$N := t \rightarrow \frac{1}{2} + \frac{1}{2}\mathrm{erf}\left(\frac{1}{2}\,t\sqrt{2}\right)$$

$$JJ(x, z) = \frac{1}{6}\left(N(t)\, z - D(N)(t)\sqrt{30}\right)\Big|_{t = \frac{6\,x - z}{\sqrt{30}}}$$

$$\text{true}$$

That says: one wants to solve the following equation for x, if z is given

```
> 'eval( N(t)*z / D(N)(t), t = (6*x-z)/sqrt(30))' = sqrt(30);
```

$$\left.\frac{N(t)\,z}{D(N)(t)}\right|_{t=\frac{6x-z}{\sqrt{30}}}=\sqrt{30}$$

From the posted solution 'OIF' one knows: for z towards 0 the desired x explodes to infinity. And so does t = (6*x-z)/sqrt(30). But then numerically the cum. normal equals 1.0 and for that asymptotic case one can solve for x in an explicit way to have an initial guess for the solver:

```
> 'eval( ``(1)*z / D(N)(t), t = (6*x-z)/sqrt(30))' = sqrt(30);
  expand(%): combine(%, exp):

  [solve(%, x)]:

  x=%[1];
  simplify(%) assuming 0 < z: evalf(%):

  initial_x:=unapply(rhs(%), z);

  plot([OIF, initial_x], 0 .. 1, 0 .. 5, color=[red,blue], labels=[z,x], title = "fsolve and initial
  guess");
```

$$\left.\frac{(1)\,z}{D(N)(t)}\right|_{t=\frac{6x-z}{\sqrt{30}}}=\sqrt{30}$$

$$x=\frac{z}{6}+\frac{1}{6}\sqrt{30}\sqrt{\ln\left(\frac{15}{\pi z^2}\right)}$$

$$\text{initial\_x} := z \to 0.166666666666667\,z+0.912870929175278\sqrt{1.56332031525281-2.\ln(z)}$$



Comparing the numerical solution and the guess is fine. Note, that this way the troubles in the 'steep' region towards z ~ 0 are done.

For the 2nd step switch to log to solve the equation, a kind of recipe for flat or steep regions. Just dare. Here there is no further need.

```
> 'eval(ln(N(t)*z) - ln(D(N)(t)) = ln(sqrt(30)), t = (6*x-z)/sqrt(30))';
  simplify(%) assuming 0<z,0<x: combine(%):


  g:=unapply(%, x, z);
```

$$(\ln(N(t)\,z) - \ln(D(N)(t)) = \ln(\sqrt{30}\,))\Big|_{t = \frac{6\,x - z}{\sqrt{30}}}$$

$$g := (x, z) \rightarrow \frac{3}{5}\,x^2 - \frac{1}{5}\,x\,z + \frac{1}{60}\,z^2 + \ln\left(\frac{1}{2}\sqrt{2}\left(1 + \operatorname{erf}\left(\frac{1}{30}\,(6\,x - z)\sqrt{15}\right)\right)\right) + \ln(z\sqrt{\pi}\,) = \frac{1}{2}\ln(30)$$

Now everything is ready to refine the way and from that the desired integral is computed in about 0.5 seconds:

```
> s:= z -> fsolve(g(x,z), x=initial_x(z));
```

$$s := z \rightarrow \text{fsolve}(g(x, z), x = \text{initial\_x}(z))$$

```
> st:=time():
  'Int(z-> s(z)*PDF(Q,z), 0 .. 1)';
  ``=evalf(%);
  `time used`[sec] = time() - st;
```

$$\text{Int}(z \rightarrow s(z)\ \text{Statistics:-PDF}(Q, z), 0\,..\,1)$$

$$= 0.282975811112917$$

$$\text{time used}_{sec} = 0.546$$